Scilab Home page | Wiki | Bug tracker | Forge | Mailing list archives | ATOMS | File exchange
Change language to: Français - Português - 日本語 - Русский

See the recommended documentation of this function

# damp

Natural frequencies and damping factors.

### Calling Sequence

[wn,z] = damp(sys)
[wn,z] = damp(P [,dt])
[wn,z] = damp(R [,dt])

### Parameters

sys

A linear dynamical system (see syslin).

P

An array of polynomials.

R

An array of real or complex floating point numbers.

dt

A non negative scalar, with default value 0.

wn

vector of floating point numbers in increasing order: the natural pulsation in rad/s.

z

vector of floating point numbers: the damping factors.

### Description

The denominator second order continuous time transfer function with complex poles can be written as s^2 + 2*z*wn*s + wn^2 where z is the damping factor and wn the natural pulsation.

If sys is a continuous time system, [wn,z] = damp(sys) returns in wn the natural pulsation (in rad/s) and in z the damping factors of the poles of the linear dynamical system sys. The wn and z arrays are ordered according to the increasing pulsation order.

If sys is a discrete time system [wn,z] = damp(sys) returns in wn the natural pulsation (in rad/s) and in z the damping factors of the continuous time equivalent poles of sys. The wn and z arrays are ordered according to the increasing pulsation order.

[wn,z] = damp(P) returns in wn the natural pulsation (in rad/s) and in z the damping factors of the set of roots of the polynomials stored in the P array. If dt is given and non 0, the roots are first converted to their continuous time equivalents. The wn and z arrays are ordered according to the increasing pulsation order.

[wn,z] = damp(R) returns in wn the natural pulsation (in rad/s) and in z the damping factors of the set of roots stored in the R array. If dt is given and non 0, the roots are first converted to their continuous time equivalents. wn(i) and z(i) are the the natural pulsation and damping factor of R(i).

### Examples

s = %s;
num = 22801 + 4406.18*s + 382.37*s^2 + 21.02*s^3 + s^4;
den = 22952.25 + 4117.77*s + 490.63*s^2 + 33.06*s^3 + s^4
h = syslin('c', num/den);
[wn,z] = damp(h)

The following example illustrates the effect of the damping factor on the frequency response of a second order system.

s = %s;
wn = 1;
clf();
Z = [0.95 0.7 0.5 0.3 0.13 0.0001];
for k=1:size(Z,'*')
z = Z(k)
H = syslin('c', 1 + 5*s + 10*s^2, s^2 + 2*z*wn*s + wn^2);
gainplot(H, 0.01, 1)
p = gce();
p = p.children;
p.foreground = k;
end
title("$\frac{1+5 s+10 s^2}{\omega_n^2+2\omega_n\xi s+s^2}, \quad \omega_n=1$")
legend('$\xi = '+string(Z)+'$')
plot(wn/(2*%pi)*[1 1], [0 70], 'r') // Natural pulsation Computing the natural pulsations and daping ratio for a set of roots:

[wn,z] = damp((1:5)+%i)
{"ft_lang_label":"__label__en","ft_lang_prob":0.59287477,"math_prob":0.99895436,"size":4071,"snap":"2021-21-2021-25","text_gpt3_token_len":1474,"char_repetition_ratio":0.1155643,"word_repetition_ratio":0.1591512,"special_character_ratio":0.32498157,"punctuation_ratio":0.10560626,"nsfw_num_words":0,"has_unicode_error":false,"math_prob_llama3":0.9989534,"pos_list":[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18],"im_url_duplicate_count":[null,1,null,1,null,1,null,1,null,1,null,1,null,1,null,1,null,1,null],"WARC_HEADER":"{\"WARC-Type\":\"response\",\"WARC-Date\":\"2021-05-10T03:21:17Z\",\"WARC-Record-ID\":\"<urn:uuid:9adb9a56-ed23-4647-b860-e6e1d1a56431>\",\"Content-Length\":\"41790\",\"Content-Type\":\"application/http; msgtype=response\",\"WARC-Warcinfo-ID\":\"<urn:uuid:0563e1fa-3403-4beb-b1ca-17c29a9a41ab>\",\"WARC-Concurrent-To\":\"<urn:uuid:b6fd4ecf-76f0-4aa2-9efb-a6e6d6f1dcc0>\",\"WARC-IP-Address\":\"176.9.3.186\",\"WARC-Target-URI\":\"https://help.scilab.org/docs/5.5.1/en_US/damp.html\",\"WARC-Payload-Digest\":\"sha1:RDLCZZGDUFWX6QZGOIZZ7GT3GJZWW6CJ\",\"WARC-Block-Digest\":\"sha1:RCUVRR4NXKVCHLPRBIBD57VB3LB5QQTG\",\"WARC-Identified-Payload-Type\":\"application/xhtml+xml\",\"warc_filename\":\"/cc_download/warc_2021/CC-MAIN-2021-21/CC-MAIN-2021-21_segments_1620243989030.65_warc_CC-MAIN-20210510003422-20210510033422-00130.warc.gz\"}"}